9

GAINAZALEN ADIERAZPEN GRAFIKOA

9.1. Aurrez definitutako figurak

▼ Graphics3D[ ] funtzioa

★ Esferak

{1, 1, 1} zentrudun eta 2 erradiodun esfera

```
Clear["Global`*"]
Graphics3D[Sphere[{0, 0, 0}, 2]]
```



Hainbat esfera

```
Graphics3D[{Sphere[{0, 0, 0}, 1], Sphere[{3, 3, 0}, 2]}]
```



★ **Zilindroak**

```
Graphics3D[Cylinder[]]
```



Jatorritik {1, 1, 1} punturainoko r=2 erradioko zilindroa

```
Graphics3D[{Opacity[0.8], Cylinder[{{0, 0, 0}, {1, 1, 1}}, 2]},
 Axes → True, AxesLabel → {"x", "y", "z"}]
```



★ **Konoak**

```
Graphics3D[Cone[]]
```



```
Graphics3D[Cone[{{0, 0, 0}, {1, 1, 1}}, 1 / 2]]
```



9.2. Parametrizatutako kurba eta gainazalak

▼ ParametricPlot3D[ ] funtzioa

★ ParametricPlot3D[{funtzioax ,funtzioay, funtzioaz},{u, umin, umax}]

Funtzio honek parametrizatutako kurba bat espazioan adierazteko balio du, u aldagaiak umin eta umax tartean balioak hartzen dituelarik

```
c = ParametricPlot3D[{Sin[u], Cos[u], u / 10},
  {u, 0, 20}, PlotStyle → Directive[Darker[Green], Thickness[0.01]]]
```



★ ParametricPlot3D[{funtzioax ,funtzioay, funtzioaz},{u, umin, umax},{v, vmin, vmax}]

Funtzio honek parametrizatutako gainazal bat espazioan adierazteko balio du, u eta v aldagaiek umin eta umax eta vmin eta vmax tarteetan balioak hartzen dituztelarik

```
s = ParametricPlot3D[{v * Sin[u], v * Cos[u], u / 10}, {u, 0, 4 Pi}, {v, 0, 1},
  PlotStyle → Directive[Purple, Opacity[0.5]], BoxRatios → Automatic, Mesh → 5]
```




```
Show[{s, c}, Axes → False]
```



▼ ParametricPlot funtzioaren aukerak

```
ParametricPlot3D[ {Cos[ ϕ] Sin[θ], Sin[ϕ] Sin[θ], Cos[θ]},
 {ϕ, 0, 2 π}, {θ, 0, π}, PlotPoints → 100, Mesh → None,
 ColorFunction → Function[{x, y, z, ϕ, θ}, Hue[Sin[6 ϕ] Sin[6 θ]]],
 ColorFunctionScaling → False]
```



▼ Koordenatu esferikoak

Gainazal baten adierazpena egiteko koordenatu esferikoak erabil daitezke:

x(u,v)= r Sin[u] Sin[v]

y(u,v)= r Cos[u] Sin[v]

z(u,v)= r Cos[v]

★ **Esfera**

```
ParametricPlot3D[ {2 Sin[u] Sin[v], 2 Cos[u] Sin[v], 2 Cos[v]},
 {u, -π, π}, {v, -π, π}, PlotStyle -> Opacity[0.5]]
```



▼ Koordenatu zilindrikoak

Gainazal baten adierazpena egiteko koordenatu zilindrikoak erabil daitezke:

x(u,v)= r Cos[u]

y(u,v)= r Sin[u]

z(u,v)= v

★ **Zilindroa**

r=2 erradioko eta z=0 baliotik z=1 baliorako altuera duen zilindroa

```
ParametricPlot3D[{2 Cos[u], 2 Sin[u], v}, {u, 0, 2 Pi}, {v, 0, 1},
 Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Pink, Blue]]
```



★ **Konoa**

```
ParametricPlot3D[{v Cos[u], v Sin[u], 2 v}, {u, 0, 2 Pi}, {v, 0, 1},
 Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Purple, Green]]
```



★ **Elipsoidea**

```
ParametricPlot3D[{2 Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]}, {v, 0, Pi}, {u, 0, 2 Pi},
 Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Orange, Yellow]]
```



★ **Biraketa toroa**

```
ParametricPlot3D[{(2 + Cos[v]) Cos[u], (2 + Cos[v]) Sin[u], Sin[v]}, {u, 0, 2 Pi},
 {v, 0, 2 Pi}, Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Purple, Green]]
```



★ **Möbius-en banda**

```
ParametricPlot3D[{Cos[t] (3 + r Cos[t / 2]), Sin[t] (3 + r Cos[t / 2]), r Sin[t / 2]},
 {r, -1, 1}, {t, 0, 2 Pi}, Mesh → {5, 10}, PlotStyle → FaceForm[Orange, Yellow]]
```



9.3. Espazioko eremuak

▼ RegionPlot3D[ ] funtzioa

Adierazitako funtzioaren barruko aldea irudikatzeko balio du. Adibidez, paraboloide honen barruko aldea honela irudika daiteke:

```
RegionPlot3D[x^2 + y^2 < z, {x, -2, 2}, {y, -2, 2}, {z, 0, 2}]
```



```
RegionPlot3D[x^2 + y^2 < z, {x, 0, 2}, {y, -2, 2}, {z, 0, 2}]
```



▼ Función ContourPlot3D[ ]

Funtzioaren gainazala irudikatzeko balio du. Adibidez, paraboloide honen gainazala honela irudika daiteke:

```
ContourPlot3D[x^2 + y^2 == z, {x, -2, 2}, {y, -2, 2}, {z, 0, 2}]
```



★ **Kuboa**

```
RegionPlot3D[-1 ≤ x ≤ 1 && -1 ≤ y ≤ 1 && -1 ≤ z ≤ 1,
 {x, -2, 2}, {y, -2, 2}, {z, -2, 2}, Mesh → None, PlotPoints → 50]
```



★ **Azal esferikoa**

```
RegionPlot3D[1 ≤ x^2 + y^2 + z^2 ≤ 3, {x, -2, 2},
 {y, -2, 2}, {z, -2, 0}, Mesh → None, PlotPoints → 50]
```



★ **Beste zenbait eremu**

```
RegionPlot3D[x^2 + y^2 + z^2 ≤ 4 && y^2 + z^2 ≤ 3, {x, -2, 2}, {y, -2, 2}, {z, -2, 2}]
```



```
RegionPlot3D[x^2 + z^2 ≤ 4 && y^2 + z^2 ≤ 3, {x, -2, 2}, {y, -2, 2}, {z, -2, 2}]
```



```
Plot3D[10 - x^2 - 2 y^2, {x, -3, 3}, {y, -3, 3},
 RegionFunction → Function[{x, y, z}, 8 < 2 x^2 + 3 y^2 < 10]]
```



9.4. Biraketa gainazalak

★ Astroidea

```
ParametricPlot[{Cos[t]^3, Sin[t]^3}, {t, 0, 2 π}, AspectRatio → Automatic,
 PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]},
 PlotLabel → Style["ASTROIDEA", Bold, Blue, 14]]
```



```
RevolutionPlot3D[{Cos[t]^3, Sin[t]^3}, {t, -Pi / 2, Pi / 2}, AspectRatio → Automatic]
```



★ **Biraketa paraboloidea**

```
Clear["Global`*"]
splot1 = ParametricPlot[{t, 0.5 t^2}, {t, -4, 4}, AxesOrigin → {0, 0},
  AspectRatio → Automatic, PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]},
  PlotLabel → Style["PARABOLA", Bold, Blue, 14]]
elipsea[t_, a_, b_, c_, d_] = {a * Sin[t], b * Cos[t]} + {c, d};
splot2 = ParametricPlot[
  Evaluate[{elipsea[t, 3, 1, 0, 0] + {0, 4.5}}, {t, 0, 2 Pi}, AspectRatio → Automatic]]
Show[splot1, splot2]
```

PARABOLA



PARABOLA



```
RevolutionPlot3D[{t, 0.5 t^2}, {t, 0, 2 Pi}]
```



★ **Biraketa toroa**

```
splot3 = ParametricPlot[{{-2 - Cos[t], Sin[t]}, {2 + Cos[t], Sin[t]}},
  {t, 0, 2 π}, PlotRange -> {{-3.1, 3.1}, {-1.5, 1.5}}, AxesOrigin → {0, 0},
  AspectRatio → Automatic, PlotStyle → {{RGBColor[0.7, 0.3, 0.2], Thickness[0.01]},
    {RGBColor[0.7, 0.3, 0.2], Dashing[0.01]}}, PlotLabel →
   Style["(0,2) zentrudun eta r=1 erradiodun zirkunferentzia", Bold, Blue, 14]]
elipsea[t_, a_, b_, c_, d_] = {a * Sin[t], b * Cos[t]} + {c, d};
splot4 = ParametricPlot[
  Evaluate[{elipsea[t, 3, 0.5, 0, 0] + {0, 0}}, {t, 0, 2 Pi}, AspectRatio → Automatic]]
Show[splot3, splot4]
```

(0,2) zentrudun eta r=1 erradiodun zirkunferentzia



(0,2) zentrudun eta r=1 erradiodun zirkunferentzia



```
ParametricPlot3D[{(2 + Cos[v]) Cos[u], (2 + Cos[v]) Sin[u], Sin[v]}, {u, 0, 2 Pi},
 {v, 0, 2 Pi}, Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Purple, Green]]
```